

Metodo di Eliminazione di Gauss versione 1

Enrico Bertolazzi

– Carica la libreria Linear Algebra per la gestione di vettori e Matrici

```
> restart ;  
with(LinearAlgebra):
```

– Procedura upper_solve

```
> upper_solve := proc (A,b)  
  local j, k, n, sol ;  
  n := RowDimension(A) ;  
  sol := Vector(n) ;  
  for k from n by -1 to 1 do  
    sol[k] := b[k] ;  
    for j from k+1 to n do  
      sol[k] := sol[k] - A[k,j]*sol[j] ;  
    end do ;  
    sol[k] := sol[k] / A[k,k] ;  
  end do ;  
  return sol ;  
end proc :
```

– Procedura gauss

Con pivoting parziale e scambio effettivo delle righe.

Notate che la matrice A e il vettore b vengono MODIFICATI!

```
> gauss := proc (A,b)  
  local i, j, k, imax, n, bf ;  
  n := RowDimension(A) ;  
  
  for i from 1 to n-1 do  
    # pivoting parziale  
    imax := i ;  
    for k from i+1 to n do  
      if abs(A[k,i]) > abs(A[imax,i]) then  
        imax := k ;  
      end if ;  
    end do ;  
  end do ;
```

```

# scambio delle righe
for k from 1 to n do
  bf      := A[i,k] ;
  A[i,k]  := A[imax,k] ;
  A[imax,k] := bf ;
end do ;
bf      := b[i] ;
b[i]    := b[imax] ;
b[imax] := bf ;

# eliminazione
for k from i+1 to n do
  bf := A[k,i] / A[i,i] ;
  for j from i+1 to n do
    A[k,j] := A[k,j] - bf * A[i,j] ;
  end do ;
  b[k] := b[k] - bf * b[i] ;
end do ;
end do ;

# operazioni di ritorno
return upper_solve(A,b) ;
end proc ;

```

- Esempio d'uso

Definisco la matrice e il vettore dei termini noti:

```

> A := Transpose(<<4,-1,1>|<1,4,1>|<-1,-1,4>>) ;
b := <1,2,3> ;

```

$$A := \begin{bmatrix} 4 & -1 & 1 \\ 1 & 4 & 1 \\ -1 & -1 & 4 \end{bmatrix}$$

$$b := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

```

> gauss(A,b) ;

```

$$\begin{bmatrix} \frac{2}{19} \\ \frac{5}{19} \\ \frac{16}{19} \end{bmatrix}$$

Notate che la matrice A e il vettore b sono stati modificati!!!!

```
> print (A,b) ;
```

$$\begin{bmatrix} 4 & -1 & 1 \\ 1 & \frac{17}{4} & \frac{3}{4} \\ -1 & \frac{-5}{4} & \frac{76}{17} \end{bmatrix}, \begin{bmatrix} 1 \\ \frac{7}{4} \\ \frac{64}{17} \end{bmatrix}$$

```
>
```